

```

/** Create library*/
libname health "\\wolftech.ad.ncsu.edu\cos\stat\Redirect\twedwar2\Documents";

/** 1. Read in aspirin data.
    subj: subject id
    y1: prothrombin time before
    y2: prothrombin time after taking aspirin
    D: difference */
data health.blood;
    input subj y1 y2 D;
    datalines;
1 12.3 12.0 0.3
2 12.0 12.3 -0.3
3 12.0 12.5 -0.5
4 13.0 12.0 1.0
5 13.0 13.0 0.0
6 12.5 12.5 0.0
7 11.3 10.3 1.0
8 11.8 11.3 0.5
9 11.5 11.5 0
10 11.0 11.5 -0.5
11 11.0 11.0 0
12 11.3 11.5 -0.2
    ;
run;

/** Null: is that time is unaffected by aspirin (y1-y2) = 0
    Alternative hypothesis: is that y1-y2 != 0
    With a test statistic of 0.74 we fail to reject the null
    and conclude that aspirin has no effect.*/
proc ttest data = health.blood;
    paired y1 * y2;
run;

/** Read in data in long form
    Add aspirin variable that is 0 if before treatment, 1 after taking aspirin.*/
data health.blood2;
    input subject @ ;
    /** This part creates two records from each line and add the aspirin variable.*/
    do k = 1 to 2 ;
        input time @ ;
        if k = 1 then aspirin = 0;
        else aspirin = 1;
        output;
    end;
    drop k ;
    datalines;
1 12.3 12.0 0.3
2 12.0 12.3 -0.3
3 12.0 12.5 -0.5
4 13.0 12.0 1.0
5 13.0 13.0 0.0
6 12.5 12.5 0.0
7 11.3 10.3 1.0
8 11.8 11.3 0.5

```

```

9 11.5 11.5 0
10 11.0 11.5 -0.5
11 11.0 11.0 0
12 11.3 11.5 -0.2
;
run;

/** Fit model blocked by subject.
    Effect of aspirin not significant.
    Subject effect significant. */
proc glm data = health.blood2;
    class subject aspirin;
    model time = subject aspirin;
run;quit;

/** 2. Cholesterol data*/
/** Read in data; 28 patients.
    7 in each group.
    Cohort: the group, defined by the age and the sex*/
data health.cholest;
    input cohort $ age $ sex $ @;
    retain cohort age sex ;
    do k = 1 to 7;
        input chol @;
        output;
    end;
    drop k;
    datalines;
I younger women 221 213 202 183 185 197 162
II younger men 271 192 189 209 227 236 142
III older women 262 193 224 201 161 178 265
IV older men 192 253 248 278 232 267 289
;
run;

/** Fit model, multiple comparisons of groups, construct contrasts of interest. */
proc glm data = health.cholest;
    class cohort;
    model chol = cohort / clparm;

    /** Multiple comparisons of groups*/
    means cohort /t cldiff;

    /** Test some specific contrasts separately*/
    estimate 'between ages' cohort 1 1 -1 -1;
    estimate 'between sexes' cohort 1 -1 1 -1 ;
    estimate 'between ages for women' cohort 1 0 -1 0;
    estimate 'between ages for men' cohort 0 1 0 -1;
    estimate 'between sexes for younger' cohort 1 -1 0 0 ;
    estimate 'between sexes for older' cohort 0 0 1 -1;
    estimate 'diff between ages same for men and women' cohort 1 -1 -1 1;
run;quit;

/** 3 Beehive data*/

```

```

/** trmt: treatment combination
temp: temperature of the hive
sucrose: amount of sugar in food soruce
energy: energy expended by bees*/
data bees;
input trmt temp sucrose @ ;
do k = 1 to 3 ;
input energy @ ;
output;
end;
drop k ;
datalines;
1 20 20 3.1 3.7 4.7
2 20 40 5.5 6.7 7.3
3 20 60 7.9 9.2 9.3
4 30 20 6 6.9 7.5
5 30 40 11.5 12.9 13.4
6 30 60 17.5 15.8 14.7
7 40 20 7.7 8.3 9.5
8 40 40 15.7 14.3 15.9
9 40 60 19.1 18.0 19.9
;
run;

/** Fit model with temperature and sucrose as variables.
Construct 4 contrasts*/
proc glm data=bees;
class temp sucrose;
model energy=temp|sucrose;
lsmeans temp*sucrose / slice=sucrose;
contrast 'contrast 1' temp*sucrose 0 -1 1 0 1 -1 0 0 0;
contrast 'contrast 2' temp*sucrose 0 .5 -.5 0 .5 -.5 0 -1 1;
contrast 'contrast 3' temp*sucrose -1 .5 .5 1 -.5 -.5 0 0 0;
contrast 'contrast 4' temp*sucrose .5 -.25 -.25 .5 -.25 -.25 -1 .5 .5 ;
run;quit;

/** 4 Root Data */
/** Read in data
comb: the treatment combination
pre: the level of pre-treatment
var: the plant variety
length: growth of roots in centimeters, this is the response*/
data root;
input comb $ pre var length @@; /**@@ because several records on the same line*/
datalines;
A 1 1 11 B 1 2 26 C 1 3 17 D 1 4 8 A 1 1 5 B 1 2 13 C 1 3 30 D 1 4 5 A 1 1 7 B 1 2 15 C 1 3 21 D 1 4 5
E 2 1 15 F 2 2 20 G 2 3 15 H 2 4 15 E 2 1 17 F 2 2 21 G 2 3 29 H 2 4 12 E 2 1 4 F 2 2 20 G 2 3 28 H 2
I 3 1 3 J 3 2 5 K 3 3 6 L 3 4 10 I 3 1 1 J 3 2 4 K 3 3 3 L 3 4 10 I 3 1 6 J 3 2 4 K 3 3 4 L 3 4 5
;
run ;

/** Fit a one way model with trt combination */
PROC GLM data = root;
class comb;

```

```
model length = comb;  
run;quit;  
  
/** Fit a model to test for the strength of the factorial effects*/  
PROC GLM data = root;  
class pre var ;  
model length = pre|var;  
lsmeans pre*var / slice=pre;  
run;quit;
```